query: #defaultView:Timeline
  SELECT DISTINCT ?t ?ppLabel ?valueLabel ?rrLabel  ?image
  WHERE
  {
    {wd:Q9711 ?prop ?t FILTER (datatype(?t)=xsd:dateTime) .
     ?value wikibase:directClaim ?prop. }
    UNION
    {
     wd:Q9711 ?q ?statement  FILTER regex (STR(?q),"prop/P") .
     ?statement ?p ?t FILTER regex (STR(?p), "prop/qualifier/P") FILTER (datatype(?t)=xsd:dateTime) .
     ?statement ?r ?value FILTER regex (STR(?r),"prop/statement/") .
     BIND(IRI(CONCAT(CONCAT(SUBSTR(STR(?r),1,29),"direct/"),SUBSTR(STR(?r),40))) as ?rprop)  
     ?rr wikibase:directClaim ?rprop .  
     BIND(IRI(CONCAT(CONCAT(SUBSTR(STR(?p),1,29),"direct/"),SUBSTR(STR(?p),40))) as ?pprop)  
      ?pp  wikibase:directClaim ?pprop .
      OPTIONAL {?value wdt:P18 ?image}
    }
    UNION
    {
     wd:Q9711 wdt:P800 ?value .
     ?value ?prop ?t FILTER (datatype(?t)=xsd:dateTime) .
     ?rr wikibase:directClaim wdt:P800 .
     ?pp wikibase:directClaim ?prop.
      OPTIONAL {?value wdt:P18 ?image}
    }
    SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
  }
  ORDER BY ?t